
 

Chapter 10:  Assembler For Dummies  
  

In which we learn that working with dummies can be the smart thing to do.  
  

A DSECT or Dummy Section provides a programmer with the ability to describe the layout of 
a storage area without reserving virtual storage. The DSECT layout can then be used to reference 
any storage area which is addressable by the program. Think of the DSECT as a template or pattern, 
which can be “dropped” on any area of storage. Once the DSECT is positioned, the symbolic names in 
the DSECT can be used to extract or change data in the underlying storage area. DSECTs provide 
programmers with the ability to assign symbols to storage areas dynamically – a powerful tool. 
  

Let’s start by creating a DSECT called CUSTOMER that describes three storage fields. 
 
CUSTOMER  DSECT  
FNAME     DS     CL10  
LNAME     DS     CL10 
BALANCE   DS     PL5  
 

The CUSTOMER DSECT is created by coding the DSECT directive, which indicates the beginning of 
the dummy section. The general format for this directive is listed below.  
  

NAME     DSECT   COMMENT 
  
The pattern for the DSECT is created using a series of DS directives to describe a collection of fields 
in storage:  FNAME, LNAME, and BALANCE. You can also use DC’s inside a DSECT, but remember that 
a DSECT acquires addressability dynamically. There is no storage area automatically associated 
with the DSECT. As a result, if you code a DC inside a DSECT, no storage is initialized. The 
assembler will consider the length of the constant in any calculations, but the constant is not created 
in memory. You should keep in mind that the length associated with the DSECT name is 1 
(L’CUSTOMER = 1). If you need the length of all the fields in the DSECT, you will need to create a 
symbol inside the DSECT that references everything. For example, CUST references the entire 
DSECT. 
 

CUSTOMER  DSECT  
CUST      DS    0CL25 
FNAME     DS     CL10  
LNAME     DS     CL10 
BALANCE   DS     PL5  

 
The end of a DSECT is indicated by the beginning of a CSECT or another DSECT. We will 

come back to this point later. 
 
After creating a DSECT, we establish addressability for its fields in a two-step process: 
 
1)  A register is associated with the DSECT name by coding a USING directive,  
 
2)  The register is loaded with the address of a storage area.  

 
  



 
In this way, the DSECT can be used to access any available storage. Let’s use this idea to extract the 
first two entries below.                      
 
MAIN     CSECT  
         ...  
         USING CUSTOMER,R7    ASSOCIATE R7 WITH THE CUSTOMER DSECT (STEP 1) 
* R7 NOW CONTROLS THE DSECT 
         LA   R7,CUST1        LOAD R7 TO DROP THE DSECT ON CUST1 (STEP 2) 
         MVC  NAME1,FNAME     FNAME REFERENCES FRED 
         MVC  NAME2,LNAME     LNAME REFERENCES SMITH 
         ZAP  BALOUT,BALANCE  BALANCE REFERENCES 432.98 
* GO AFTER CUST2 
         LA   R7,CUST2        LOAD R7 TO DROP THE DSECT ON CUST2 (STEP 2) 
         MVC  NAME1,FNAME     FNAME REFERENCES SUSAN 
         MVC  NAME2,LNAME     LNAME REFERENCES JONES 
         ZAP  BALOUT,BALANCE  BALANCE REFERENCES 123.45  
         ...  
* FIRST CUSTOMER 
CUST1    DS  0CL35 
         DC   CL10’FRED’  
         DC   CL20’SMITH’  
         DC   PL5’432.98’  
* SECOND CUSTOMER 
CUST2    DS  0CL35 
         DC   CL10’SUSAN’ 
         DC   CL20’JONES’ 
         DC   PL5’123.45’ 
         ... 
NAME1    DS   CL10 
NAME2    DS   CL10 
BALOUT   DS   PL5 
 

After defining the dummy section, it must be associated with an available register in a 
USING directive. In the example code above, the statement USING CUSTOMER,R7 associates register 
seven with the CUSTOMER dummy section. Addressability is established when the register is loaded 
with the address of a CUST1. At that point, FNAME contains the value “FRED      ”, LNAME contains 
the value “SMITH     ”, and BALANCE contains a packed decimal 432.98. Loading the register has 
“dropped” the DSECT on the storage area whose address is contained in the register, and it has 
given us access to the subfields of CUST1. We repeat the technique by loading R7 with the address of 
CUST2. After the second load address, FNAME contains the value “SUSAN     ”, LNAME contains  
“JONES     ”, and BALANCE contains a packed decimal 123.45. 

  
Consider the power of the above technique. There were no symbolic names associated with 

the subfields of the first or second customers above. By creating a DSECT, we can dynamically 
assign symbols to storage areas and access the data the areas contain. DSECTs have many uses in 
assembly language, including array processing, locate mode I/O, and data structure creation and 
manipulation. Many sophisticated assembler programs use DSECTS. Let’s consider the use of 
DSECTs in array processing as a way of starting our study of DSECTs.  
  
  



 
Loading an Array With Data  
  

In the following example, we will load a storage area with data that is read from a file. The 
first 100 records in the file will be read and stored in the table. We assume the file has at least 100 
records. The FILEIN DCB is not specified, and we will use a DSECT similar to the previous example.  
  

CUSTOMER DSECT  
FNAME    DS     CL10   
LNAME    DS     CL10 
BALANCE  DS     PL5   
CRECLEN  EQU    *-CUSTOMER       LENGTH OF CUSTOMER ENTRY 
 
MAIN CSECT  
         ...  
         USING CUSTOMER,R7       R7 CONTROLS DSECT POSITION 
         LA    R7,TABLE          POSITION DSECT AT TOP OF TABLE 
         L     R4,NOITEMSF       LOAD # OF TABLE ENTRIES INTO R4  
LOOP     EQU *  
         GET   FILEIN,RECIN      READ A RECORD  
         MVC   FNAME,FNAMEIN     MOVE DATA ...   
         MVC   LNAME,LNAMEIN     ... FROM RECIN ...  
         ZAP   BALANCE,BALIN     ... TO THE TABLE  
         LA    R7,CRECLEN(R0,R7) BUMP THE DSECT TO NEXT TABLE ITEM 
         BCT   R4,LOOP           BRANCH BACK IF MORE RECS  
         ...  
NOITEMS  EQU   100               TABLE WILL HAVE ROOM FOR 100 ITEMS 
NOITEMSF DC    A(NOITEMS)        NO OF ITEMS AS A FULLWORD 
TABLE    DS    (NOITEMS)CL(CRECLEN)  AREA FOR 100 CUSTOMER RECORDS  
 
RECIN    DS   0CL80 
FNAMEIN  DS    CL10 
LNAMEIN  DS    CL10 
BALIN    DS    PL5 
         DS    CL55  

       
The table's address is placed in R7 by the first LA instruction. This has the effect of 

positioning the DSECT at the beginning of the table area since R7 is associated with CUSTOMER in 
the USING statement. A record is read, and the subfields are moved into the table by two MVCs and 
a ZAP. The most interesting line of code is LA  R7,CRECLEN(R0,R7).We are loading an address 
into R7. The effective address of Operand two is computing by adding CRECLEN and the contents 
of R7. (R0 is ignored in the calculation.) This has the effect of adding 25 to the contents of R7 and 
putting the sum back in R7, “bumping” the DSECT forward 25 bytes. We loop back, read another 
record, and move it to the next area in the table. We continue in this way until 100 records are 
loaded, using R4 to control the loop. 
  

To move the DSECT to the next empty table entry, we coded the following line,  
  

LA R7,CRECLEN(R0,R7) BUMP THE DSECT  
  
This type of statement is often used in DSECT processing, so we consider it in some detail. Even 
though the statement uses explicit addressing – a technique we usually try to avoid - here, the 
technique is regarded as an acceptable practice. The effect of the instruction is to compute the second 



operand address explicitly from the base register (R7), index register (R0), and displacement 
(CRECLEN). When R0 is specified as an index register, that component of the address computation 
is ignored. The “effective” address consists of the contents of R7 plus the displacement of 25 
represented by CRECLEN. Since R7 contains an address of a table entry, adding 25 to R7 will create 
the address of the next table entry. The DSECT has been “bumped” down to the next table entry.  
  

Finally, the BCT instruction is used for loop control and ensures that each entry in the table 
is filled.   
  
  
 Where Do DSECTS Live? 
 

The assembler provides great freedom in the placement of DSECTs.  If you are just 
beginning to experiment with DSECTs, you might start by placing your DSECTs above the 
main CSECT like those in the example below: 

 
CUSTA   DSECT 
A       DS    CL3 
B       DS    CL4 
CUSTB   DSECT 
C       DS    CL4 
D       DS    CL10 
MAIN    CSECT    
        ... 
        END   MAIN 
 
The CUSTA DSECT ends when the CUSTB DSECT begins. The CUSTB DSECT ends 

when the MAIN CSECT begins. The MAIN CSECT ends with the END statement. 
 
Another option is to put your DSECTS at the end like this: 
 
MAIN     CSECT 
        ... 
CUSTA   DSECT 
A       DS    CL3 
B       DS    CL4 
C       DS    CL10 
CUSTB   DSECT 
D       DS    CL4 
E       DS    CL10 
MAIN    CSECT    
        ... 
        END   MAIN 

 
 After coding the two DSECTS, we restart the main control section before coding END. 
  

  



Like CSECTs, DSECTs can also be stopped and started again. The assembler will 
put the correct pieces together in the right order, so the following example is equivalent to 
the example above. 

   
MAIN     CSECT 
        ... 
CUSTA   DSECT 
A       DS    CL3 
B       DS    CL4 
CUSTB   DSECT 
D       DS    CL4 
CUSTA   DSECT 
C       DS    CL10 
CUSTB   DSECT 
E       DS    CL10 
MAIN    CSECT    
        ... 
        END   MAIN 
 

The CUSTA DSECT consists of three fields: A, B, and C. The CUSTB DSECT consists of two 
fields: D and E. The idea of mixing CSECTs and DSECTS is a bit advanced for a beginning 
text, but you should be aware that it’s possible. 

 

The Magic of DSECTs 

How does this DSECT magic work? When the assembler processes a DSECT, it 
resets the location counter for the DSECT to 0 to compute displacements to fields inside. 
 

The example below lists a DSECT and the corresponding location counter values. 
 
LOC 
            CUSTOMER DSECT  
000000      FNAME    DS     CL10   
00000A      LNAME    DS     CL10 
000014      BALANCE  DS     PL5   
 
If we code USING   R7,CUSTOMER , the assembler will use seven as the base 

register along with the appropriate displacement above when creating base/displacement 
addresses.  For example, the base/displacement address for BALANCE will be 7014.  
Everything will work out as long as R7 contains the appropriate runtime address. When 
you are using DSECTs, the USING statement's associated register controls everything.  
This is backward from the way normal symbols work, where we fix a base address in the 
register and vary the displacement to get to the field we are addressing. With DSECTs, the 
offsets are fixed, and we vary the contents of the base register. Clever!  This has an 
important implication:  If you can make a register point to something you want, you can use 
a DSECT to grab it. (Of course, it has to be something in your region, otherwise you will get 
a S0C4.) 
 


